Introducción a la Programación con la Calculadora Gráfica *fx*-9750G PLUS

Ángel Aguirre Pérez, Profesor de Matemáticas del I.E.S. "Benedicto Nieto", de Pola de Lena (Asturias)

En este artículo vamos a explorar la potencia de la calculadora gráfica *fx*-9750G PLUS en el cálculo integral. El programa en esta entrega, nos va permitir resolver el ya clásico problema de la cabra. (1*)Un enunciado de este problema podría ser el siguiente:

"Una cabra está atada, mediante una cuerda, a un poste de la periferia de un cercado circular de un decámetro de radio. Si sólo puede pastar la mitad de la superficie del campo, ¿qué longitud debe tener la cuerda?"

La forma matemática del problema puede ser: ¿Qué radio ha de tener un circunferencia con centro en el punto (0,1) para que seccione en dos partes iguales el círculo de radio unidad centrado en el origen de coordenadas?

El círculo unitario se descompone en dos partes una lúnula (parte inferior) y una superficie ovalada (parte superior).



Es indudable que la longitud de la cuerda (el radio de la circunferencia), r, ha de estar comprendido entre los valores 1 y $\sqrt{2}$. Si la cuerda tiene una longitud de 1, la lúnula tiene el área sombreada de la figura de abajo en exceso.



Por otra parte si la longitud de la cuerda fuese $\sqrt{2}$, sería ahora el oval el que excedería en área a la lúnula; tal como muestra, de nuevo, la figura de la abajo.



Una longitud intermedia (entre 1 y $\sqrt{2}$), para la cuerda que ata la cabra, parece ser la solución. Lo que la cabra come de más en la parte central del círculo debe ser compensado con lo que deja de comer en los extremos.



En este momento nos damos cuenta de que el problema tiene simetría y que queda reducido a una igualdad de superficies entre los dos recintos sombreados.

Nuestros conocimientos en cálculo integral nos permiten reducir el problema a la siguiente condición:

$$\int_0^a \left(1-\sqrt{r^2-x^2}\right)dx+\int_a^1 \sqrt{1-x^2}\,dx=0$$

donde a es la abscisa del punto de corte de la circunferencia de radio unidad, $x^2+y^2=1$, y la curva, también circunferencia $x^2+\left(y-1\right)^2=r^2$. La continuación habitual sería hallar el valor de la abscisa del punto de corte en función de la longitud de la cuerda r, hallar las primitivas de ambas integrales, aplicar la regla de Barrow para evaluar las integrales definidas dejando las expresiones en función de r y resolver la ecuación planteada. Nuestra estrategia explotará la posibilidad que la calculadora gráfica nos ofrece para evaluar integrales definidas y su capacidad de definir funciones a trozos.

$$\int_0^1 f(x)\,dx$$

Se transforma la ecuación original en:

$$f(x)=\begin{cases}1-\sqrt{r^2-x^2} & si \quad 0\le x\le a\\ \sqrt{1-x^2} & si \quad a<x\le 1\end{cases}$$

El problema es hallar el valor de r que anula la función $área(r)=\int_0^1 f(x)dx$.

```
======CABRA    ======
Sci 0↵
1→I↵
√2→S↵
Locate 1,1,"PROBLEMA
DE LA CABRA"↵
Locate 1,2,"CALCULAND
O..."↵
```

Comenzamos tomando dos valores de la abscisa donde dicha integral toma valores de signo opuesto. En nuestro caso estos valores serán almacenados en las variables I (extremo inferior) y S (extremos superior) y serán 1 y $\sqrt{2}$ respectivamente. Previamente incluimos "Sci 0" porque queremos que la información numérica que salga por pantalla sea en notación científica mostrando 10 dígitos significativos. También advertimos mediante sentencias "Locate" que empezamos los cálculos para resolver el problema de la cabra.

A continuación empezamos un bucle en el que en cada iteración vamos a acercarnos gradualmente a la solución.

```
Do↵
0.5×(I+S)→R↵
0.5×R×√(4-R²)→A↵
∫((1-√(R²-X²))×(X≤A)+
√(1-X²)×(X>A),0,1→T↵
```

Tomamos como primera aproximación de r (R en el programa) al punto medio de los extremos superior e inferior. Para ese valor de la cuerda calculamos la abscisa del punto de corte con la expresión $\frac{R}{2}\sqrt{4-R^2}$ y lo guardamos en la variable A e inmediatamente se evalúa la integral que se almacena en la variable T.

Obsérvese la manera de describir una función a trozos utilizando los operadores <, ≤, > y ≥.

```
If T<0↵
Then R→S↵
Else R→I↵
IfEnd↵
ClrText↵
Locate 1,1,"*********
************"↵
Locate 1,2,"MEJOR VAL
OR R Y AREA"↵
Locate 1,3,"HASTA EL
MOMENTO SON"↵
Locate 1,4,R↵
Locate 1,5,T↵
Locate 1,6,"*********
************"↵
LpWhile (Abs T>1E-12)
↵
```

Si el área es nula, el problema ha terminado y éste el valor de R buscado. Si el valor de T es negativo estamos ante un nuevo extremo superior menor que el anterior y guardamos su valor en S.

Si por el contrario el valor de T es positivo estamos ante un nuevo extremo inferior mayor que el anterior y guardamos su valor en I. De cualquier modo la solución queda confinada en un nuevo intervalo (I, S) de menor longitud a cada iteración. Mostramos nuestros cálculos hasta el momento por pantalla y repetimos el proceso hasta que la integral alcanza un valor muy pequeño razonable (en nuestro caso por debajo de 10^{-12}). Esto se hace así porque en la evaluación por métodos numéricos de la integral, rara vez se obtiene un valor nulo para ésta debido a que la precisión está limitada en los cálculos intermedios.

```
ClrText↵
ClrGraph↵
ViewWindow -2.1,2.1,1,-1.05,1.05,1↵
r=Type↵
"1"→r1↵
G SelOn 1↵
Y=Type↵
"1-√(R²-X²)"→Y2↵
G SelOn 2↵
DrawGraph↵
Graph ∫√(1-X²)×(X>A)+(1-√(R²-X²))×(X≤A),0,1◢
```

Una vez alcanzada la precisión requerida del cálculo aprovechamos las capacidades gráficas de nuestra calculadora para mostrar la solución. En primer lugar limpiamos la pantalla de cualquier texto y gráfico anterior. Ajustamos el tamaño de la ventana que vamos a mostrar y dibujamos dos curvas: una circunferencia de radio unidad en polares, $r=1$, y la función en explícitas $y=1-\sqrt{r^2-x^2}$. Para comprobar el resultado, calculamos la integral gráfica de la función definida a trozos *f(x)*. El mando de salida, ◢ , asegura que visualicemos la representación gráfica.

Para acabar el programa muestra la solución mediante unas sentencias "Locate".

```
Locate 1,2,"**********************"↵
Locate 1,3,"LA SOLUCION ES R ="↵
Locate 1,4,R↵
Locate 1,5,"**********************"↵
```

La ejecución del programa CABRA puede apreciarse en las siguientes pantallas:





(1*) (El problema de la cabra, Alberto Bagazgoitia, Sigma nº 21).